

FUNCTIONAL PEARL

Building a consensus: A rectangle covering problem

RICHARD S. BIRD

*Computing Laboratory, Oxford University
Wolfson Building, Parks Road, Oxford, OX1 3QD, UK
(e-mail: bird@comlab.ox.ac.uk)*

1 Introduction

The other day, over a very pleasant lunch in the restaurant of Oxford's recently renovated Ashmolean Museum,¹ Oege de Moor gave me a problem about rectangles. The problem is explained more fully later, but roughly speaking one is given a finite set of rectangles RS and a rectangle R completely covered by RS . The task is to construct a single rectangle covering R among the elements of a larger set of rectangles associated with RS , called the *saturation* of RS . The saturation of RS is the closure of RS under so-called *consensus* operations, a term coined in (Quine, 1959), in which two rectangles are combined in two distinct ways to form new rectangles. The rectangle problem is a simplified version of containment-checking, a crucial component in a type inference algorithm for Datalog programs (Schäfer & de Moor, 2010). In the Schäfer-de Moor algorithm the problem is generalised to cubes in n -space rather than rectangles in two-space, the components of each cube are given by propositional formulae rather than by intervals on the real line, and certain equality and inhabitation constraints are taken into account. Oege felt that the central proof, Lemma 15 in (Schäfer & de Moor, 2010), deserved to be simplified so he posed the rectangle problem as a special case. This pearl was composed in response to the challenge.

2 The problem

Consider Figure 1, which depicts four rectangles A , B , C and D , covering another rectangle X framed with dotted lines. It happens in this example that none of the four rectangles can be removed without uncovering some part of X , but that is not essential to the problem. Next, consider two binary operations, \oplus_0 and \oplus_1 , on rectangles. Assuming A and B overlap, the operation $A \oplus_0 B$ forms a new rectangle by taking the union of the horizontal intervals defining A and B and the intersection of the vertical intervals. The operation $A \oplus_1 B$ takes the intersection of the horizontals

¹ Well worth a visit if you are ever in Oxford, especially as entry to the museum is free.




Fig. 1. Four rectangles covering the dotted rectangle.




Fig. 2. The dotted rectangles are $A \oplus_0 D$ and $A \oplus_1 D$.

and the union of the verticals. These two operations are the *consensus* operations on A and B . For example, Figure 2 shows the two consensus rectangles for the rectangles A and D of Figure 1. The consensus operations also apply when the two rectangles do not overlap, though the result is either the empty rectangle or two rectangles.

Formally we define a rectangle A to be a pair (X_A, Y_A) of intervals on the real line.² The points in A are the elements in the cartesian product $X_A \times Y_A$. The two consensus operations are defined by

$$\begin{aligned}
 A \oplus_0 B &= (X_A \cup X_B, Y_A \cap Y_B) \\
 A \oplus_1 B &= (X_A \cap X_B, Y_A \cup Y_B)
 \end{aligned}$$

Both \oplus_0 and \oplus_1 are associative, commutative and idempotent; moreover each operation distributes over the other. These properties follow from the same properties of union and intersection. By definition, the *saturation* of a set of rectangles RS is the least set XS containing RS and containing $A \oplus_0 B$ and $A \oplus_1 B$ for all A and B in XS . Assuming RS covers R , our task is to construct a single rectangle X in the saturation of RS that covers R . We do so by building a *consensus expression* that describes X . A consensus expression is a binary tree whose leaves are labelled with suitable names for the rectangles in RS and whose nodes are labelled with one or other of the two consensus operations. Failure to construct a consensus means that the rectangles RS do not cover R . To appreciate the problem, the reader is invited to construct a consensus expression over the four rectangles A, B, C, D of

² The assumption that X_A and Y_A are intervals is considered further in Section 4.1.

Figure 1 that covers the fifth rectangle X . The answer is given at the beginning of the following section.

The rectangle problem admits an immediate generalisation to n -cubes, which are cubes in n -dimensional space. An n -cube A is defined by an n -tuple of sets $[X_A, Y_A, \dots, Z_A]$. The points in A are the elements of $X_A \times Y_A \times \dots \times Z_A$, and there are n consensus operations \oplus_j for $0 \leq j < n$. The value of $A \oplus_j B$ is the cube in which the k th component for $k \neq j$ is the intersection of the k th components of A and B for $0 \leq k < n$, and the j th component is the union of the j th components of A and B .

Finally, let us briefly relate the above consensus operations to those of same name in the Quine–McCluskey method for finding prime implicants. Consider a boolean formula in conjunctive normal form over n propositional variables. For example, taking $n = 3$ we might have the expression

$$a'b'c + abc' + a'bc + abc$$

in which the propositional variables are a , b and c , and dashed letters denote negation. By definition, the consensus of $a'b'c$ and $a'bc$ is the formula $a'c$, equivalently $a'(b + b')c$. If we interpret a term $a'b'c$ as a cube $[a', b', c]$ in which a' and b' are the complements of a and b with respect to some fixed universe, then the consensus of $a'b'c$ and $a'bc$ is just $[a', b', c] \oplus_1 [a', b, c]$.

3 Proof and first construction

One way of covering X in Figure 1 is with the rectangle

$$(B \oplus_0 C) \oplus_1 (A \oplus_0 C) \oplus_1 (A \oplus_0 D)$$

but there are many other expressions that also do the job.

Now, generalising from rectangles to n -cubes, how shall we prove that there is a consensus expression over a set of n -cubes XS that covers a given cube X if and only if XS does? To answer this question let us first make a simplifying assumption, namely that each component of each cube is a *finite set of integers*. We will see in Section 4 that this assumption entails no loss of generality. Given that each component has finite size, we can prove the theorem by induction on the size of X , defined to be the product of the sizes of the components of X . A cube X of size 1 has singleton components and so contains just a single point. A single point can clearly be covered by a single cube in XS if and only if XS covers X . This establishes the base case. Otherwise, by way of induction, assume that all cubes of size less than s can be covered by a single cube in the saturation of XS . Let X be a cube of size $s > 1$, so at least one component of X is a non-singleton set. Say the k th component is not a singleton. Split X into two cubes Y and Z by splitting the k th component into two smaller sets, leaving the other components unchanged. Each of Y and Z has size smaller than s , so by induction there exist cubes A and B in the saturation of XS that cover Y and Z , respectively. Now X is covered by $A \oplus_k B$.

The proof easily converts into a constructive procedure for computing the required expression. Turning to Haskell, we first declare

```
type Cube      = [Component]
type Component = [Int]
```

A cube is a list of n components and each component is a finite, possibly empty list of integers in strictly increasing order. We allow empty components because the result of a consensus operation might be the empty cube. Each cube represents a set of points in n -space:

```
type Point      = [Int]
points          :: Cube → [Point]
points [xs]     = [[x | x ← xs]
points (xs : xss) = [x : ys | x ← xs, ys ← points xss]
```

A set xcs of cubes covers a cube xc if for each point of xc there is some member of xcs that contains it:

```
covers          :: [Cube] → Cube → Bool
covers xcs xc   = and [contains xcs p | p ← points xc]
contains xcs p  = or [p ∈ points xc | xc ← xcs]
```

Any set of cubes trivially covers the empty cube because the empty cube contains no points.

The operation \oplus_k is implemented as a function *consensus* k :

```
consensus       :: Int → Cube → Cube → Cube
consensus k xc yc = zipWith cap xc1 yc1 ++ [cup xs ys] ++ zipWith cap xc2 yc2
                 where (xc1, xs : xc2) = splitAt k xc
                       (yc1, ys : yc2) = splitAt k yc
```

Definitions of *cup* and *cap*, the union and intersection functions on ordered lists, are omitted.

Next, we declare a suitable type for consensus expressions:

```
data Exp      = Val Label | Op Int Exp Exp
type Label    = Int
```

The evaluation function *eval* is defined by

```
eval           :: [Cube] → Exp → Cube
eval xcs (Val j) = xcs !! j
eval xcs (Op k u v) = consensus k (eval xcs u) (eval xcs v)
```

Cubes are labelled by their positions in the given list of cubes.

The function *cover0* is a straightforward implementation of the induction proof (later on, we consider other versions, *cover1*, *cover2*, and so on):

```

cover0 :: [Cube] → Cube → Exp
cover0 xcs xc
    = if null yss
      then Val (head [j | (j, yc) ← zip [0..] xcs, covers [yc] xc])
      else Op (length xss) (cover0 xcs yc) (cover0 xcs zc)
      where
        (xss, yss) = span single xc
        yc         = xss ++ [[head (head yss)]] ++ tail yss
        zc         = xss ++ [tail (head yss)] ++ tail yss
    
```

The test *single* returns *True* on a singleton list and *False* otherwise. The expression *span single xc* splits the components of *xc* into two lists, *xss* and *yss*, where *xss* consists of singleton sets only and *yss* is either empty or begins with a non-singleton set. If *yss* is empty, then *xc* consists of a single point. By construction,

$$\text{covers } xcs \text{ } xc \Rightarrow \text{covers } [eval \text{ } xcs \text{ } (cover0 \text{ } xcs \text{ } xc)] \text{ } xc$$

For example, *cover0 xcs [[1, 2, 3], [1, 2], [1, 2]]* produces the expression

$$\begin{aligned}
 & ((111 \oplus_2 112) \oplus_1 (121 \oplus_2 122)) \oplus_0 \\
 & ((211 \oplus_2 212) \oplus_1 (221 \oplus_2 222)) \oplus_0 \\
 & ((321 \oplus_2 312) \oplus_1 (311 \oplus_2 322))
 \end{aligned}$$

in which the term *abc* represents the label of the first cube in *xcs* that covers the point $[a, b, c]$.

4 Digitization

We now show why we can assume that the components of cubes are finite sets of integers. An interval on the real line is defined by a pair (a, b) of numbers representing the set of real numbers x in the range $a \leq x < b$. The length of (a, b) is $b - a$, so a pair (a, a) denotes an empty interval. There is no loss in generality in requiring a and b to be integers, but, even so, there is an infinite number of points in each nonempty interval.

The solution is to digitize real intervals as integer intervals. Consider for example the three rectangles of Figure 3. Suppose their component intervals are as follows:

$$A = [(0, 50), (40, 80)], \quad B = [(30, 70), (10, 60)], \quad C = [(40, 80), (0, 50)]$$

The dotted lines shown in the figure extend each horizontal and vertical line to the x and y axes and partition the axes into five horizontal segments

$$H_0 = (0, 30), \quad H_1 = (30, 40), \quad H_2 = (40, 50), \quad H_3 = (50, 70), \quad H_4 = (60, 80)$$

and five vertical segments

$$V_0 = (0, 10), \quad V_1 = (10, 40), \quad V_2 = (40, 50), \quad V_3 = (50, 60), \quad V_4 = (60, 80)$$




Fig. 3. Dividing the axes into segments.

By construction, the components of A , B and C can each be expressed as the union of segments. For example, $X_A = H_0 \cup H_1 \cup H_2$ and $Y_A = V_2 \cup V_3 \cup V_4$. Going one step further, we can now represent each rectangle by the lists of integer labels of the horizontal and vertical segments that make up its coordinates. That gives new rectangles

$$A = [[0 .. 2], [2 .. 4]], B = [[1 .. 3], [1 .. 3]], C = [[2 .. 4], [0 .. 2]]$$

whose components are intervals of integers. Moreover, containment-checking on the original problem returns the same result as on the digitized version. However, digitization is possible only because each cube is assumed to have components that are intervals.

To implement digitization we first declare the types

```
type RealCube = [Interval]
type Interval = (Int, Int)
```

We digitize each list of corresponding components separately:

```
digitize :: [RealCube] → [Cube]
digitize = transpose · map digitizeRow · transpose
```

The list of cubes is transposed to bring each corresponding component into the same row. Each row is then digitized and the result is transposed back to give a new list of cubes. To implement *digitizeRow*, we first sort the boundaries of each interval into strictly increasing order, discarding duplicates:

```
boundaries :: [Interval] → [Int]
boundaries = sort · concatMap (λ(a,b) → [a,b])
```

The definition of *sort* is omitted. The function *mksegs* constructs a new list of intervals out of the boundaries:

```
mksegs :: [Int] → [Interval]
mksegs xs = zip xs (tail xs)
```

Each new interval is then encoded as a component:

```
encode :: [Interval] → Interval → Component
encode row (x,y) = [j .. k]
where j = head [j | (j,(a,b)) ← jabs, a = x]
      k = head [j | (j,(a,b)) ← jabs, b = y]
      jabs = zip [0 .. ] row
```

That gives

$$\text{digitizeRow row} = \text{map}(\text{encode}(\text{mksegs}(\text{boundaries row})))\text{row}$$

4.1 Propositional components

We mentioned briefly in the Introduction that in Datalog containment-checking the components of rectangles were propositional formulae rather than intervals. For example, consider the rectangles

$$\begin{aligned} A &= (a + c, a + bc) \\ B &= (a, b' + c) \\ C &= (bc, a + b'c) \\ D &= (a + bc, b') \\ X &= (a + bc, a + b') \end{aligned}$$

over the propositional variables a , b and c . The rectangle X is covered by A , B , C and D in the sense that

$$\begin{aligned} (a + bc) &\Rightarrow (a + c) + a + bc + (a + bc) \\ a + b' &\Rightarrow (a + bc) + (b' + c) + (a + b'c) + b' \end{aligned}$$

We can convert this version of the rectangle problem into an integer version by coding formulae using the truth table for a , b and c :

	abc		abc
0	000	4	100
1	001	5	101
2	010	6	110
3	011	7	111

Using the labels of the entries in this table, the five rectangles above translate to the integer rectangles

$$\begin{aligned} A &= ([1, 3, 4, 5, 6, 7], [3, 4, 5, 6, 7]) \\ B &= ([4, 5, 6, 7], [0, 1, 3, 4, 5]) \\ C &= ([3, 7], [1, 4, 5, 6, 7]) \\ D &= ([3, 4, 5, 6, 7], [0, 1, 4, 5]) \\ X &= ([3, 4, 5, 6, 7], [0, 1, 4, 5, 6, 7]) \end{aligned}$$

Of course, the components are not intervals, but `cover0` still works because it did not assume interval components.

5 Optimisation

Let us now return to cubes whose components are lists of integers. The trouble with `cover0.xcs xc` is that each point of xc is associated with a cube in xcs and these cubes are then assembled into a potentially very large consensus expression. Because adjacent points of xc may well be associated with the same cube, and each consensus operation is idempotent, there is room for improvement both in speed and expression size.

One way to reduce the size of the final consensus expression is to reduce the size of the cubes using digitization. For example, consider again the rectangles of Figure 3, this time as intervals of integers:

$$A = [[0 .. 49], [40 .. 79]], B = [[30 .. 69], [10 .. 59]], C = [[40 .. 79], [0 .. 49]]$$

Each of these rectangles is of size 2000, which is quite large. But digitization converts the problem into one whose rectangles each have size 9, a substantial reduction. The function *cover1* is defined by

$$\begin{aligned} \text{cover1} & \quad :: [Cube] \rightarrow Cube \rightarrow Exp \\ \text{cover1 } xcs \ xc & = \text{cover0 } ycs \ yc \\ & \quad \textbf{where } yc : ycs = \text{digitize } (\text{map } \text{toPair}) (xc : xcs) \\ \text{toPair } xs & = (\text{head } xs, \text{last } xs + 1) \end{aligned}$$

The function *cover1*, which is a valid optimisation only if the components of the cubes are contiguous intervals of integers, may reduce the size of the consensus expression but not necessarily minimize it.

In search of an algorithm that does minimize expression size, let us first recast *cover0* in iterative form. To this end, consider the function *expand k* that splits a cube into s_k subcubes, where s_k is the size of the k th component:

$$\begin{aligned} \text{expand} & \quad :: Int \rightarrow Cube \rightarrow [Cube] \\ \text{expand } kxc & = [xss \oplus [[y]] \oplus yss \mid y \leftarrow ys] \\ & \quad \textbf{where } (xss, ys : yss) = \text{splitAt } k \ xc \end{aligned}$$

Using *expand* we can define

$$\begin{aligned} \text{cover2 } xcs \ xc & = \text{pcover } 0 \ xc \\ \textbf{where} & \\ \text{pcover } kxc & = \textbf{if } k = n \\ & \quad \textbf{then } \text{Val } (\text{head } [j \mid (j, yc) \leftarrow \text{zip } [0 ..] \ xcs, \text{covers } [yc] \ xc]) \\ & \quad \textbf{else } \text{foldr1 } (Op \ k) (\text{map } (\text{pcover } (k+1))) (\text{expand } k \ xc) \\ n & = \text{length } xc \end{aligned}$$

We omit the proof that $\text{cover0} = \text{cover2}$. The function *cover2* is faster than *cover0*, though not dramatically so. However, *cover2* gets us, quite literally, to the nub of the problem. Let \oplus be some associative, commutative and idempotent operation. The critical fact about \oplus is that

$$\text{foldr1 } (\oplus) = \text{foldr1 } (\oplus) \cdot \text{nub}$$

where *nub* is the standard Haskell library function that removes duplicates from a list. The value of $\text{foldr1 } (\oplus) \ xs$ depends only on the *set* of elements in *xs* and not on the order of the elements or on duplications.

We could minimize the size of the final consensus expression simply by inserting a *nub* after $\text{foldr1 } (Op \ k)$ in the definition of *pcover*, but *nub* uses a quadratic number of equality tests and each equality test compares two consensus expressions, so the new version, *cover3* say, may well be slower than *cover2*.

Here is a better way to remove duplicates. The idea is to convert the target cube xc to an n -dimensional array of points, then to replace each point by the label of the first cube in xcs which covers it. This array of labels is then pruned, dimension by dimension, by removing all duplicates in each row, then all duplicate rows, all duplicate planes, and so on. To represent an n -dimensional array in Haskell we need trees:

$$\mathbf{data} \text{ Tree } a = \text{Leaf } a \mid \text{Fork } [\text{Tree } a]$$

The function *convert* converts an n -cube into a list of trees of points:

$$\begin{aligned} \text{convert} &:: \text{Cube} \rightarrow [\text{Tree Point}] \\ \text{convert } [xs] &= [\text{Leaf } [x] \mid x \leftarrow xs] \\ \text{convert } (xs : xss) &= [\text{Fork } (\text{map } (\text{cons } x) \text{ ts}) \mid x \leftarrow xs] \\ &\quad \mathbf{where} \text{ ts} = \text{convert } xss \\ \text{cons } x (\text{Leaf } xs) &= \text{Leaf } (x : xs) \\ \text{cons } x (\text{Fork } ts) &= \text{Fork } (\text{map } (\text{cons } x) \text{ ts}) \end{aligned}$$

The function *covering* xcs labels a tree of points with the first cube in xcs that covers each point:

$$\begin{aligned} \text{covering} &:: [\text{Cube}] \rightarrow \text{Cube} \rightarrow [\text{Tree Label}] \\ \text{covering } xcs \text{ xc} &= \text{map label } (\text{convert } xc) \\ \mathbf{where} & \\ \text{label } (\text{Leaf } xs) &= \text{Leaf } (\text{lookup } xs) \\ \text{label } (\text{Fork } ts) &= \text{Fork } (\text{map label } ts) \\ \text{lookup } x &= \text{head } [j \mid (j, xc) \leftarrow \text{zip } [0..] \text{ xcs}, \text{member } x \text{ xc}] \end{aligned}$$

The function *member* is defined by

$$\text{member } xxc = \text{and } [a \leq y \wedge y < b \mid ((a, b), y) \leftarrow \text{zip } (\text{map toPair } xc) \text{ x}]$$

The function *prune* removes duplicate labels:

$$\begin{aligned} \text{prune} &:: [\text{Tree Label}] \rightarrow [\text{Tree Label}] \\ \text{prune } ts @(\text{Leaf } _ : _) &= \text{remdups } ts \\ \text{prune } ts &= \text{remdups } [\text{Fork } (\text{prune } vs) \mid \text{Fork } vs \leftarrow ts] \\ \text{remdups} &:: \text{Eq } a \Rightarrow [a] \rightarrow [a] \\ \text{remdups} &= \text{foldr } \text{op} [] \\ \mathbf{where} \text{ opx } [] &= [x] \\ \text{opx } (y : ys) &= \mathbf{if } x = y \mathbf{then } y : ys \mathbf{else } x : y : ys \end{aligned}$$

The function *remdups* removes adjacent duplicates only, but that is sufficient because in the cubes problem all duplicates will occur together. The function *lookup* always returns the first cube in xcs that covers the point x and that fact guarantees that duplicate cubes occur together.

Now we can define our final version:

```

cover4                :: [Cube] → Cube → Exp
cover4 xcs xc         = mkexp 0 (prune (covering ycs yc))
  where yc : ycs = digitize (map (map toPair) (xc : xcs))

mkexp k ts@(Leaf _ : _) = foldr1 (Op k) [Val x | Leaf x ← ts]
mkexp k ts               = foldr1 (Op k) [mkexp (k+1) vs | Fork vs ← ts]

```

6 Experimental results

Using `ghci` we carried out a brief comparison of the five versions of `cover`. Recall that `cover0` was the divide and conquer version, `cover1` the version that reduced problem size by digitization, `cover2` was the linearised form of `cover0`, `cover3` used the Haskell `nub` function, and `cover4` the super-duper version that used both digitization and a tailored implementation for removing duplicates. The results, using three sample sets of cubes, are summarised in the table below. Times are in seconds, and size refers to the size of the consensus expression. The set `xcs0` is a set of rectangles of total area 525, the set `xcs1` a set of rectangles of total area 161, and `xcs2` a set of four-dimensional cubes of area 8097.

	xcs0		xcs1		xcs2	
	time	size	time	size	time	size
<code>cover0</code>	0.05	241	1.64	503	38.97	1295
<code>cover1</code>	0.00	49	0.02	31	0.06	23
<code>cover2</code>	0.05	241	1.63	503	38.89	1295
<code>cover3</code>	0.05	17	1.59	3	38.75	9
<code>cover4</code>	0.02	17	0.02	3	0.02	9

Clearly `cover4` was best by a long way, so it was worth the effort to obtain it.

Acknowledgments

The author thank Jeremy Gibbons and two anonymous referees for their constructive remarks and suggestions. As a result the pearl has been thoroughly overhauled.

References

- Quine, W. V. (1959) On cores and prime implicants of truth functions. *Am. Math. Mon.*, **66**(9), 755–760.
- Schäfer, M. & de Moor, O. (2010) Type inference for datalog with complex type hierarchies. *Princ. Program. Lang. 2010. ACM SIGPLAN Not.*, **45**(1), 145–156.